Q:

Adding hook on elisp function

I'm using Emacs. 
Is there any way to add hook on a function? 
Assume that there is a markdown-export function. 
It is designed to export HTML file into current directory where current working 'markdown file' exsits. 
But, I want to export HTML file into another directory.  How can I do that without modification on Emacs markdown plugin (markdown-mode.el)? 
This is markdown-mode.el's export function:
(defun markdown-export (&optional output-file)
  "Run Markdown on the current buffer, save to file, and return the filename. 
If OUTPUT-FILE is given, use that as the filename.   Otherwise, use the filename
generated by `markdown-export-file-name', which will be constructed using the
current filename, but with the extension removed and replaced with .html." 
  (interactive)
  (unless output-file
    (setq output-file (markdown-export-file-name ".html")))
  (when output-file
    (let* ((init-buf (current-buffer))
           (init-point (point))
           (init-buf-string (buffer-string))
           (output-buffer (find-file-noselect output-file))
           (output-buffer-name (buffer-name output-buffer)))
      (run-hooks 'markdown-before-export-hook)
      (markdown-standalone output-buffer-name)
      (with-current-buffer output-buffer
        (run-hooks 'markdown-after-export-hook)
        (save-buffer))
      ;; if modified, restore initial buffer
      (when (buffer-modified-p init-buf)
        (erase-buffer)
        (insert init-buf-string)
        (save-buffer)
        (goto-char init-point))
      output-file)))

=====================================================================
I have made an advice to save exported HTML at temp directory
Here is the code. 
(defadvice markdown-export (around set-temp-path-for-exported-file activate)
  (ad-set-arg 0 (format "%s/%s" "~/.emacs.d/temp-dir" (file-name-nondirectory buffer-file-name)))
  ad-do-it)

Thanks!!!!!!!!!!!!!! 

A:

In this case you do not need to hook on this function since it already accepts the filename as an argument, unfortunately it does not accept the filename when called interactively.  As a workaround you can define a simple wrapper around the function like follows
(defun my-markdown-export (&optional file)
  (interactive (list (ido-read-file-name "Export as: ")))
  (markdown-export file))

A:

The advice mechanism is a bit like having hooks for any arbitrary function, but here you have actual hooks you can use, as well as a function argument which addresses your requirement directly. 
So you can:
(a) Pass the function any arbitrary output filename. 
(b) Use the provided markdown-before-export-hook to setq whichever variables you need to (which at a glance looks like output-file, output-buffer, and output-buffer-name).